
VGA 输出彩条实验说明

1、实验简介

实验使用“小眼睛科技”公司的 FPGA 开发板以及 PMOD 转 VGA 转接板，通过 VGA

接口在显示屏屏幕上显示彩条，其中显示格式为 1024*768@60。

2、实验目的

使用 VGA 线连接开发板上的 PMOD-VGA 转接板与显示器，在 VGA 屏幕上完成彩条显

示实验。

3、实验使用模块介绍

实验使用“小眼睛科技”公司的 PMOD 转 VGA 接口模块。模块共有两个 PMOD 2x6 针连

接器（J2与 J3）。

使用模块进行管脚绑定时，请仔细观察针脚位置进行管脚的绑定。

由于 VGA输入信号为模拟量，FPGA输出信号为数字量，因此模块使用权电阻网络将数字

量转换为模拟量提供给 VGA。



4、实验原理

（1）VGA接口

VGA（Video Graphics Array）视频图形阵列，于 1978被 IBM 公司提出，是电脑显示器上应

用广泛的视频接口。VGA 接口一共 15 针孔，一共三行，每行 5 孔，其中， 5个 GND信号，3

个 RGB 彩色分量信号，1个行同步信号，1 个场同步信号，1个电源信号。

15 针脚定义如下表所示：

针脚 定义 针脚 定义

1 红基色 9 保留



2 绿基色 10 数字地

3 蓝基色 11 地址码

4 地址码 12 地址码

5 自测试 13 行同步

6 红地 14 场同步

7 绿地 15 地址码

8 蓝地

（2）VGA显示

VGA显示图像使用扫描的方式，从第一行第一个像素点开始扫描，扫描到第一行最后一个

像素点后，开始扫描第二行第一个像素点，以此类推，直到扫描到最后一行最后一个像素点，

即完成一帧图像的显示，返回第一行第一个像素点，开始第二帧图像的扫描。整体扫描路线为

Z字形。

当扫描速度足够快时，由于人眼的视觉暂留特性，就可以观看到一个完整的画面；一个画

面又称为一场或一帧，每秒钟刷新的帧数又称为帧率。



（3）VGA时序

VGA显示时像素点的颜色信号源源不断的传输，此时行、场的切换需要行场同步信号控制。

行同步信号用于标志一行的开始，同样也标志上一行的结束，行同步信号的时序如下图所

示：

场同步信号用于标志一帧或一场画面的开始，同样也标志着上一场画面的结束，场同步信

号的时序如下图所示：

整体的时序如下图所示：



上图中 Addressable 部分内容是在显示器中可看到的区域，Border可理解为显示黑边或者

显示边框，通常 Border（边框）显示黑色。行、场切换过程都是在用户感受不到的区域进行的，

这个区域就是 Blanking部分，称为消隐区间。同步信号上升沿表示新的一行/一场开始，Hsync

对应行，Vsync 对应场。

因此，VGA 刷新一帧的时间不仅与有效像素有关，还与消影、同步的时间有关。

本实验采用 1024*768@60的视频规格，详细时序参数如下：



5、实验模块设计

实验采用 1024*768@60 的视频规格，VGA 时钟为：1344*806*60=65MHz；通过视频规格

参数，在行同步位置拉高行同步信号，在场同步位置拉高场同步信号，在有效像素位置输入基

色信号。



`timescale 1ns / 1ps
`define UD #1

module vga_test(
input clk ,
input rstn ,
output reg vs_out ,
output reg hs_out ,
output reg[3:0]r_out ,
output reg[3:0]g_out ,
output reg[3:0]b_out
);

parameter X_WIDTH = 4'd12;
parameter Y_WIDTH = 4'd12;

reg de_out ;

parameter V_TOTAL = 12'd806;
parameter V_TB = 12'd0;
parameter V_BB = 12'd0;
parameter V_FP = 12'd3;
parameter V_BP = 12'd29;
parameter V_SYNC = 12'd6;
parameter V_ACT = 12'd768;
parameter H_TOTAL = 12'd1344;
parameter H_LB = 12'd0;
parameter H_RB = 12'd0;
parameter H_FP = 12'd24;
parameter H_BP = 12'd160;
parameter H_SYNC = 12'd136;
parameter H_ACT = 12'd1024;
parameter HV_OFFSET = 12'd0;

localparam H_ACT_ARRAY_0 = H_ACT/8;
localparam H_ACT_ARRAY_1 = 2* (H_ACT/8);
localparam H_ACT_ARRAY_2 = 3* (H_ACT/8);
localparam H_ACT_ARRAY_3 = 4* (H_ACT/8);
localparam H_ACT_ARRAY_4 = 5* (H_ACT/8);
localparam H_ACT_ARRAY_5 = 6* (H_ACT/8);
localparam H_ACT_ARRAY_6 = 7* (H_ACT/8);
localparam H_ACT_ARRAY_7 = 8* (H_ACT/8);

reg [X_WIDTH-1:0] h_count = 'd0;



reg [Y_WIDTH-1:0] v_count = 'd0;

reg [X_WIDTH-1:0] x_act = 'd0;
reg [Y_WIDTH-1:0] y_act = 'd0;

reg hs ;
reg vs ;
reg de ;

U_PLL PLL_PIX (
.clkout0(pix_clk), // output
.lock(lock), // output
.clkin1(clk) // input

);

/* horizontal counter */
always @(posedge pix_clk)
begin

if (!rstn)
h_count <= `UD 0;

else
begin

if (h_count < H_TOTAL - 1)
h_count <= `UD h_count + 1;

else
h_count <= `UD 0;

end
end

/* vertical counter */
always @(posedge pix_clk)
begin

if (!rstn)
v_count <= `UD 0;

else
if (h_count == H_TOTAL - 1)
begin

if (v_count == V_TOTAL - 1)
v_count <= `UD 0;

else
v_count <= `UD v_count + 1;

end
end



always @(posedge pix_clk)
begin

if (!rstn)
hs <= `UD 1'b0;

else
hs <= `UD ((h_count < H_SYNC));

end

always @(posedge pix_clk)
begin

if (!rstn)
vs <= `UD 4'b0;

else
begin

if (v_count == 0)
vs <= `UD 1'b1;

else if (v_count == V_SYNC)
vs <= `UD 1'b0;

else
vs <= `UD vs;

end
end

always @(posedge pix_clk)
begin

if (!rstn)
de <= `UD 1'b0;

else
de <= (((v_count >= V_SYNC + V_BP + V_TB) && (v_count <= V_TOTAL - V_FP

- V_BB - 1)) && ((h_count >= H_SYNC + H_BP + H_LB) && (h_count <= H_TOTAL - H_FP - H_RB
- 1)));

end

// active pixels counter output
always @(posedge pix_clk)
begin

if (!rstn)
x_act <= `UD 'd0;

else
begin
/* X coords – for a backend pattern generator */

if(h_count > (H_SYNC + H_BP + H_LB - 1'b1))
x_act <= `UD (h_count - (H_SYNC + H_BP+ H_LB));

else



x_act <= `UD 'd0;
end

end

always @(posedge pix_clk)
begin

if (!rstn)
y_act <= `UD 'd0;

else
begin

/* Y coords – for a backend pattern generator */
if(v_count > (V_SYNC + V_BP + V_TB - 1'b1))

y_act <= `UD (v_count - (V_SYNC + V_BP + V_TB));
else

y_act <= `UD 'd0;
end

end

always @(posedge pix_clk)
begin

vs_out <= `UD vs;
hs_out <= `UD hs;
de_out <= `UD de;

end

always @(posedge pix_clk)
begin

if (de)
begin

if(x_act < H_ACT_ARRAY_0)
begin

r_out <= 4'hf;
g_out <= 4'hf;
b_out <= 4'hf;

end
else if(x_act < H_ACT_ARRAY_1)
begin

r_out <= 4'hf;
g_out <= 4'hf;
b_out <= 4'h0;

end
else if(x_act < H_ACT_ARRAY_2)
begin

r_out <= 4'h0;



g_out <= 4'hf;
b_out <= 4'hf;

end
else if(x_act < H_ACT_ARRAY_3)
begin

r_out <= 4'h0;
g_out <= 4'hf;
b_out <= 4'h0;

end
else if(x_act < H_ACT_ARRAY_4)
begin

r_out <= 4'hf;
g_out <= 4'h0;
b_out <= 4'hf;

end
else if(x_act < H_ACT_ARRAY_5)
begin

r_out <= 4'hf;
g_out <= 4'h0;
b_out <= 4'h0;

end
else if(x_act < H_ACT_ARRAY_6)
begin

r_out <= 4'h0;
g_out <= 4'h0;
b_out <= 4'hf;

end
else
begin

r_out <= 4'h5;
g_out <= 4'h5;
b_out <= 4'h5;

end
end
else
begin

r_out <= 4'h0;
g_out <= 4'h0;
b_out <= 4'h0;

end
end

endmodule



6、实验现象

使用 VGA 线连接开发板上的 PMOD-VGA 转接板与显示器如下图，下载程序，可以看到显

示器显示 8条彩条。

